
Abstract Starting with an evaluator for a language, an abstract ma-
chine for the same language can be mechanically derived using success-
ive program transformations. This has relevance to studying both the
space and time properties of programs because these can be estimated
by counting transitions of the abstract machine and measuring the size of
the additional data structures needed, such as environments and stacks.
In this paper we will use this process to derive a function that accurately
counts the number of steps required to evaluate expressions in a simple
language.

1 Introduction

The problem of reasoning about intensional properties of functional programs,
such as the running time, is itself a long-running one. It is complicated by dif-
ferent evaluation strategies and sharing of expressions, meaning that some parts
of a program may not be run or only partially so. One of the issues involved in
reasoning about the amount of time a program will take to complete is what to
count as an atomic unit in evaluation, or an evaluation step.

An evaluator is usually an implementation of the denotational semantics of
the language — it evaluates the expression based on the meaning of its sub-
expressions. This level of understanding helps us reason about extensional prop-
erties of the language and makes the evaluation strategy explicit, but it doesn’t
say anything about the underlying way that the evaluation is taking place. By
contrast, an operational semantics shows us the method that is being used to
evaluate an expression, and the conventional approach is to use this to measure
the number of steps that is required. This, however, may not be very accurate
because what is usually being measured is beta-reductions, each of which may
take arbitrarily long.

It is proposed that a more realistic measure would be to count transitions
in an actual machine. What is needed is some way to relate these two concepts
together to produce a machine implementation of a language directly from the
evaluator for it, so that we can be sure that they are both using the same
evaluation method.

In the last couple of years Olivier Danvy et al have produced several pa-
pers exploring the basis of abstract machines and outlining the process of de-
riving them from evaluators [1, 2, 3]. Abstract machines can be viewed as an
implementation of the operational semantics at a low level [4]. This process
uses, in particular, two program transformation techniques: transformation in



to continuation-passing style and defunctionalization. These will be explained in
detail when they are required later.

In this paper, we will apply this process to that of accurately counting eval-
uation steps. In brief this consists of introducing a simple language in which to
write expressions, together with an evaluator and deriving a corresponding ab-
stract machine using successive program transformations. Simple step counting
is then added to the machine, by threading and incrementing a counter that
measures the number of transitions. The next stage is to do the same process
but in reverse order to get out an evaluator that additionally counts the number
of steps, directly corresponding to the number of transitions of the underlying
abstract machine. A step counting function will be calculated from this evalu-
ator and this will be used to reason about evaluation of some example functions
expressed in the language.

All the programs are written in Haskell [5], but no specific features of this lan-
guage are used, so they may be easily adapted to other functional programming
languages.

2 Language

The test language needs to be small, so that it’s easy to reason about, but
powerful enough to express functions that have interesting time behaviour. The
language chosen is the untyped lambda calculus (variables, abstraction and ap-
plication) extended with integers, addition, lists and recursion over lists in the
form of fold-right. These could have been expressed directly in lambda calculus,
for example by using peano numbers instead of introducing integers, but this
would introduce an unrealistic overhead in evaluation. A more general recursion
operator could have been introduced instead of fold-right, but for simplicity one
tailored to our data structure, lists, is sufficient. It is also hoped that using fold-
right will simplify the process of reasoning about time properties in the same
way as it has proved useful for reasoning about extensional ones [6]. The syntax
of the language chosen is expressed below using BNF.

E ::= x | λx .e | E E | Z | E + E | [ ] | E : E | foldr E E E

Using this language some common functions over lists can be expressed, such
as summing a list, both with and without an accumulator, appending two lists,
concatening a list of lists and reversing a list, again using an accumulator or not.

The primitive functions (add , foldr and :) are implemented as fully saturated,
in that they take their arguments all at once. The main reason for doing this is to
make it easier to define what a value is. If they were introduced as constants then,
for example, add 1 would be a value. This doesn’t affect what can be expressed
in the language; partial application can be expressed by using abstractions, so
an equivalent expression would be λx .add 1 x .



3 Derivation

For simplicity, we will consider evaluation using the call-by-value strategy, where
arguments to functions are evaluated before the function is performed.

The language is implemented as the Haskell data type shown below,

data Expr = Var String | Abs String Expr | App Expr Expr
| Add Expr Expr | Val Value
| Cons Expr Expr | Foldr Expr Expr Expr

data Value = Const Int | ConsV Value Value | Nil | Clo String Expr Env

The data type diffentiates between expressions and values, in particular lists
that contain unevaluated and evaluated expressions, so that there is no need to
iterate repeatedly over the list to check if each element is fully evaluated, which
would introduce an artificial evaluation overhead.

An expression is either a variable, abstraction, application of two expressions,
addition of two expressions, fold-right over an expression (where the function
and nil-case are both expressions), a list containing further expressions or a
value. A value is either an integer, list containing further values or a closure (an
abstraction paired with an environment containing bindings for all free variables
in the abstraction).

Evaluation is performed using an environment that is passed in to the eval-
uator function and is used to look up what a variable is bound to. This avoids
having to substitute in expressions for variables, which is complicated by the
need to deal with avoiding name-capture. Under call-by-value evaluation argu-
ments are evaluated before function application, so variables will be bound to a
value. The environment is represented as a list of pairs,

type Env = [(String ,Value)]

The intermediate programs are written very concretely - the environment and
step counter are threaded through the program - instead of introducing more
structure, such as using monads. This makes it more obvious what is going on at
a low-level, in that there are no hidden steps happening, but it can be verbose, so
the derivations are shown for just a subset of the language: integers and addition.

3.1 Evaluator

An evaluator is a representation of the denotational semantics of the language.
It is compositional, in that an expression is described as the meaning of its parts,
and evaluation order isn’t specified, it will depend on the implementing language.
By contrast an abstract machine is a term rewriting system that represents a one-
step transition function. However it differs from the reflexive transitive closure
of a single step transition semantics (usually written as ∗→) because the next
transition of the machine is entirely determined by the value currently being
examined and the current state [4]. The implementation of this is a first-order
tail-recursive program.



The starting evaluator is given below,

eval :: Expr → Env → Value
eval (Val v) env = v
eval (Var x ) env = fromJust (lookup x env)
eval (Abs x e) env = Clo x e env
eval (App f e) env = let Clo x e ′ env ′ = eval f env

v = eval e env
in eval e ′ ((x , v) : env ′)

eval (Add x y) env = let Const m = eval x env
Const n = eval y env

in Const (m + n)
eval (Cons x xs) env = ConsV (eval x env) (eval xs env)
eval (Foldr f v xs) env = case eval xs env of

Nil → eval v env
ConsV z zs → let f ′ = eval f env

x = eval (Foldr (Val f ′) v (Val zs)) env
in eval (App (App (Val f ′) (Val z )) (Val x )) [ ]

Values are already evaluated, so they are just returned. Variables are evaluated
by returning the value the variable is bound to in the environment. Under call-by-
value evaluation, the arguments to functions are evaluated before the function
is performed, so no further evaluation is required. An application App f e is
evaluated by first evaluating f to an abstraction then evaluating the body of the
abstraction with the environment extended with the variable bound to the the
value that e evaluates to.

Addition is performed by first evaluating both sides to an integer and then
adding them together. This will give another integer result, so does not need to
be further evaluated. Evaluating a cons consists of evaluating the first element
and the tail and then re-assembling them to make an evaluated list.

Evaluation of the Foldr case could be specified in different ways. The com-
pletely call-by-value way would be to evaluate the arguments in left to right
order, so that the first two arguments are evaluated before the list argument.
However, for the Nil list argument case, the function argument to Foldr is evalu-
ated even though it is not required. The approach in the evaluator is to evaluate
the list argument first to allow pattern matching and then evaluate the other
arguments depending on what the list evaluated to. So when the list evaluates to
Nil only the second argument to Foldr is evaluated. The justication not to use
the purely call-by-value way is that it would introduce some artificial behaviour
of the Foldr function. When the lambda-calculus is extended with a conditional
function, for example, it is not implemented to expand both branches under
call-by-value evaluation, but in the way that it behaves in practise: evaluating
the condition and then one branch depending on the value of the condition.

Evaluating an expression is performed by passing the expression to the eval
function along with an empty environment, eval e [ ].



4 Derivation

The derivation of the step counting function will be shown in detail for a subset of
the language, for conciseness. We will consider expressions consisting of integers
and addition, with values being integers,

data Expr = Add Expr Expr | Val Value
type Value = Int

An environment is not required for this subset of the language, so it will be left
out. The starting evaluator is then simply,

eval :: Expr → Value
eval (Val v) = v
eval (Add x y) = eval x + eval y

4.1 Tail-recursive Evaluator

Our aim is to turn the evaluator in to an abstract machine; a first order, tail-
recursive program. The evaluator is already first order, but it is not tail-recursive.
It can be made so by transforming it in to continuation passing style (CPS). A
continuation is a function that represents the rest of a computation; this makes
the evaluation order of the arguments explicit, so intermediate results need to
be ordered using the continuation. A program can be transformed in to CPS
by redefining it to take an extra argument, a function which is applied to the
result of the original one. The continuation function will take an argument of
type Value, and its result is a Value,

type Con = Value → Value

The new evaluator can be calculated from the old one by using the specification,

evalTail e c = c (eval e)

By induction on e,

Case : e = Val v

evalTail (Val v) c
= { specification }

c (eval v)
= { definition of eval }

c v

Case : e = Add x y

evalTail (Add x y) c
= { specification }

c (eval (Add x y))



= { definition of eval }
c (eval x + eval y)

= { reverse beta-reduction, abstract over eval x }
(λm → c (m + eval y)) (eval x )

= { inductive assumption }
evalTail x (λm → c (m + eval y))

= { reverse beta-reduction, abstract over eval y }
evalTail x (λm → (λn → c (m + n)) (eval y))

= { inductive assumption }
evalTail x (λm → evalTail y (λn → c (m + n)))

In conclusion, we have calculated the following recursive definition,

evalTail :: Expr → Con → Value
evalTail (Val v) c = c v
evalTail (Add x y) c = evalTail x (λm →

evalTail y (λn → c (m + n)))

The semantics of the original evaluation function can be recovered by substitut-
ing in the identity function for the continuation,

eval e = evalTail e (λx → x )

4.2 Abstract Machine

The next step is to make the evaluator first order. This is done by defunction-
alising the continuation. At the moment the continuation is a function Value →
Value but the whole function space is not required; the continuation function is
only created in three different ways. Defunctionalisation is performed by look-
ing at all places where functions are made and replacing them with a new data
structure that takes as arguments any free variables required, and has an apply
function which does the same thing as the original function.

The data structure required is shown below. The names of the constructors
will become clear later.

data Cont = Top for the initial continuation (λv → v)
|AddL Cont Expr for (λ(Const m)→ evalTail y (...))
|AddR Value Cont for (λ(Const n)→ c (Const (m + n)))

By definition the apply function needs to do the same for each instance of
the continuation function,

apply Top = λv → v
apply (AddR m c) = λn → apply c (m + n)
apply (AddL c y) = λm → evalTail y (apply (AddL c m))



Using the specification, evalMachine e c = evalTail e (apply c), the function
evalMachine can be calculated by induction on e,

Case : e = Val v

evalMachine (Val v) c
= { specification }

evalTail (Val v) (apply c)
= { definition of evalTail }

apply c v

Case : e = Add x y

evalMachine (Add x y) c
= { specification }

evalTail (Add x y) (apply c)
= { definition of evalTail }

evalTail x (λm → evalTail y (λn → apply c (m + n)))
= { definition of apply }

evalTail x (λm → evalTail y (apply (AddR m c)))
= { definition of apply }

evalTail x (apply (AddL c y))
= { inductive assumption }

evalMachine x (AddL c y)

We have now calculated the following recursive function,

evalMachine :: Expr → Cont → Value
evalMachine (Val v) c = apply c v
evalMachine (Add x y) c = evalMachine x (AddL c y)

Moving the lambda abstracted terms to the left and applying the specification
in the AddL case, gives the following function,

apply :: Cont → Value → Value
apply Top v = v
apply (AddR m c) n = apply c (m + n)
apply (AddL c y) m = evalMachine y (AddL c m)

The original semantics can be recovered by passing in the equivalent of the initial
continuation, the Top constructor,

eval e = evalMachine e Top

as shown below,

eval e
= { definition of eval }

evalTail e (λv → v)
= { definition of apply }



evalTail e (apply Top)
= { specification }

evalMachine e Top

The evaluator is now an abstract machine that has two states, for eval and
apply . The reason for the constructor names can be revealed; the data structure
is the structure of evaluation contexts for the language [7]. It can alternatively
be viewed as a stack, pushing expressions still to be evaluated and values to be
saved.

4.3 Step Counting Abstract Machine

The number of time steps required to evaluate an expression is going to be
measured by counting the number of transitions of the abstract machine. The
abstract machine derived can be simply modified by adding a step count that is
incremented each time a transition, a function call to evalMachine or apply , is
made. The step count is added as an accumulator, rather than just incrementing
the count that the recursive call returns, so that it is still an abstract machine.
Then the same program transformations are performed in the reverse order to
derive an evaluator that counts steps at the evaluator level, corresponding to the
number of transitions of the abstract machine.

type Step = Int

stepMachine :: (Expr ,Step)→ Cont → (Value,Step)
stepMachine (Val v , s) c = apply ′ c (v , s + 1)
stepMachine (Add x y , s) c = stepMachine (x , s + 1) (AddL c y)

apply ′ :: Cont → (Value,Step)→ (Value,Step)
apply ′ Top (v , s) = (v , s + 1)
apply ′ (AddL c y) (m, s) = stepMachine (y , s + 1) (AddR m c)
apply ′ (AddR m c) (n, s) = apply ′ c (m + n, s + 1)

The number of steps is initialised to zero and incremented every time the machine
makes a transition. The evaluation function now returns a pair, where the first
part is the evaluated value, and the second is the number of steps taken. The
actual addition of the integers is a primitive operation and so happens in one
step no matter what the size of the arguments. The semantics of the original
evaluator can be recovered as,

eval e = fst (stepMachine e 0 Top) |

The aim now is to derive a function that counts the number of steps required to
evaluate an expression; from the specification,

steps e = snd (stepMachine (e, 0) Top)



The first step in the reverse process is to refunctionalise the representation of
the continuation. The original continuation was a function from Value → Value,
the new one will go from (Value,Step)→ (Value,Step).

type Con ′ = (Value,Step) → (Value,Step)

Again, this can be calculated by induction on e, from the following specification,

stepTail (e, s) (apply ′ c′) = evalMachine (e, s) c′

The refunctionlised version is,

stepTail :: (Expr ,Step)→ Con ′ → (Value,Step)
stepTail (Val v) s c = c (v , s + 1)
stepTail (Add x y) s c = stepTail x (s + 1) (λ(m, s ′)→

stepTail y (s ′ + 1) (λ(n, s ′′)→ c (m + n, s ′′ + 1)))

The new step counting function becomes,

steps e
= { definition of steps }

snd (stepMachine (e, 0) Top)
= { specification }

snd (stepTail (e, 0) (λ(v , s)→ apply ′ Top v s))
= { definition of apply ′ }

snd (stepTail (e, 0) ((v , s)→ (v , s + 1)))

steps e = snd (stepTail (e, 0) ((v , s)→ (v , s + 1)))

4.4 Evaluator with Accumulator

The step counting evaluator can be transformed from CPS back to direct style,
by removing the continuation, to give a function with the type,

stepAcc :: (Expr ,Step)→ (Value,Step)

This is performed by induction on e, using the specification,

c (stepAcc (e, s)) = stepTail (e, s) c

The resulting evaluator is,

stepAcc :: (Expr ,Step)→ (Value,Step)
stepAcc (Val v) s = (v , s + 1)
stepAcc (Add x y) s = let (m, s ′) = stepAcc (x , s + 1)

(n, s ′′) = stepAcc (y , s ′ + 1)
in (m + n, s ′′ + 1)



The new step counting function becomes,

steps e
= { specification }

snd (stepTail (e, 0) ((v , s)→ (v , s + 1)))
= { definition of stepTail }

snd ((λ(v , s)→ (v , s + 1)) (stepAcc (e, 0)))
= { lemma, snd }

snd (stepAcc (e, 0)) + 1

steps e = snd (stepAcc (e, 0)) + 1

4.5 Step Counting Evaluator

At the moment the step counting evaluator threads the step count as an accu-
mulator. This can be removed, by calculating a new function without one, using
the specification,

stepEval e = let (v , s ′) = stepAcc (e, s)
in (v , s ′ − s)

Again, this can be calculated by induction over the structure of the expression,
to give the new step counting evaluator,

stepEval :: Expr → (Value,Step)
stepEval (Val v) = (v , 1)
stepEval (Add x y) = let (m, s) = stepEval x

(n, s ′) = stepEval y
in (m + n, s + s ′ + 3)

The semantics of the steps function can be expressed as,

steps e = snd (stepEval e) + 1

4.6 Step Counting Function

The final stage is to calculate a standalone steps function. This will take an
expression and return the number of steps to evaluate the expression, calling the
original evaluator where the result of evaluation is required.

This can be calculated by pushing through the definition of the snd function,
to give,

steps ′ (Add x y) = steps ′ x + steps ′ y + 3
steps ′ (Val v) = 1

The steps function is then simply,

steps e = steps ′ e + 1



5 Complete Function

Adding the rest of the language and the environment, the complete steps ′ func-
tion now looks like,

steps ′ (Val v) env = 1
steps ′ (Var x ) env = 1
steps ′ (Abs x e) env = 1
steps ′ (App f e) env =

steps ′ f env + steps ′ e env + steps ′ e ′ ((x , v) : env ′) + 3
where (Clo x e ′ env ′) = eval f env

v = eval e env
steps ′ (Add x y) env = steps ′ x env + steps ′ y env + 3
steps ′ (Cons x xs) env = steps ′ x env + steps ′ xs env + 3
steps ′ (Foldr f v xs) env = steps ′ xs env + case eval xs env of

Nil → steps ′ v env + 2
ConsV y ys → steps ′ f env + steps ′ (Foldr (Val f ′) v (Val ys)) env+

steps ′ (App (App f ′ (Val y)) x ) [ ] + 4
where f ′ = Val (eval f env)

x = Val (eval (Foldr (Val f ′) v (Val ys)) env)

The derived function shows the constant overheads involved in evaluation — for
example, the number of steps needed to evaluate an addition is the sum of the
number of steps to evaluate each side plus a constant three.

We want to be able to reason about how the time usage of some example
functions depends on the size of the arguments to the function. In the case of
fold-right functions, it would be easier to reason about the time usage if it was
expressed as a function over the size of the list, rather than a recursive function.
In the steps function above the Foldr case makes a recursive call to fold the tail
of the list. This can naturally be expressed as a fold-right over the value list data
structure, defined as,

foldrVal :: (Value → b → b)→ b → Value → b
foldrVal f v Nil = v
foldrVal f v (ConsV x xs) = f x (foldrVal f v xs)

Also, if the number of steps to apply the binary function f does not depend on
the value of the arguments passed, such as adding two expressions, then this can
be expressed as a function over the length of the list argument supplied,

lengthVal = foldrVal (λ n → n + 1) 0



6 Example Functions

The derived steps function can now be used on the examples to calculate the
number of steps required in evaluating the function. QuickCheck [8] was used to
test each function produced.

6.1 Summing a List

Below are two ways to sum a list of integers expressed using the fold-right oper-
ator,

sum [ ] = 0
sum (x : xs) = x + sum xs ⇔ sum xs = foldr (+) 0 xs

sumAcc [ ] a = a
sumAcc (x : xs) a = sumAcc xs (a + x ) ⇔

sumAcc xs = foldr f id xs 0
where f x g a = g (a + x )

The first replaces each : in the list with + and the unit of addition, 0, for the
empty list case. The second has an accumulator; the fold is used to generate a
function which is applied to the identity function in the nil case and in the cons
case adds the current value to the accumulator.

Using an accumulator could potentially save on space, because additions
could be performed without having to expand the whole list first. It would be
useful to know what effect an accumulator has on the number of steps taken.

Sum

sum = Abs xs (Foldr add (Val (Const 0)) (Var xs))
add = Abs x (Abs y (Add (Var x) (Var y)))

The number of steps required to evaluate applying the sum function to a list xs
is given below,

steps (App sum (Val xs)) = 21 ∗ (lengthVal xs) + 10

The step count is a constant multiplied by the length of the list argument plus a
constant amount; it is directly proportional to the length of the list argument.

Sum with an Accumulator

sumAcc = Abs xs (App (Foldr f idExpr (Var xs)) (Val (Const 0)))
where f = Abs x (Abs g (Abs a (App (Var g) (Add (Var a) (Var x)))))

idExpr = Abs x (Var x)

The step count of the sum with an accumulator is of the same form, linear on
the length of the list, but the constant values are larger, because there is an
additional overhead in evaluating extra abstractions.

steps (App sumAcc (Val xs)) = 26 ∗ (lengthVal xs) + 15



6.2 Concatenation

Concatenating a list of lists can be defined by folding the append function over
the list,

concat xs = foldr append [ ] xs

where append is defined as,

append xs ys = foldr (λz zs → z : zs) ys xs

Append First we need to analyse the append function.

append = Abs xs (Abs ys (Foldr f (Var ys) (Var xs)))
where f = Abs z (Abs zs (Cons (Var z) (Var zs)))

The number of steps to evaluate the append function applied to two list argu-
ments is given below.

steps (App (App append (Val xs)) (Val ys)) = 21 ∗ (lengthVal xs) + 15

The function is proportional to the length of the list that the first argument
evaluates to.

Concat The step count of the concat function can now be calculated using this
function,

concat = Abs xss (Foldr append (Val Nil) (Var xss))

With the step count from the append function inlines, the function is,

steps (App concat (Val xss)) = foldrVal f 10 xss
where f ys s = 21 ∗ (lengthVal ys) + 20 + s

The number of steps required in evaluation is the sum of the steps taken to apply
the append function to each element in the list.

If the argument to concat evaluates to a list where all the list elements are of
the same length (so the number of steps taken in applying the append function
will always be constant) then this can be simplified to,

steps (App concat (Val xss)) = 20 + case xss of
Nil → 0
ConsV ys yss → (lengthVal xss) ∗ (21 ∗ (lengthVal ys))

The number of steps is now proportional to the length of the input list
multiplied by the number of steps to evaluate appending an element of the list,
which is proportional to the length of that element.



6.3 Reversing a List

Reversing a list can be expressed directly as a fold by appending the reversed
tail of the list to the head element made in to a singleton list, as shown below,

reverse [ ] = [ ]
reverse (x : xs) = reverse xs ++ [x ] ⇔

reverse xs = foldr f [ ] xs
where f x xs = xs ++ [x ]

The function can also be expressed using an accumulator,

fastreverse [ ] a = a
fastreverse (x : xs) a = fastreverse xs (x : a) ⇔

fastreverse xs = foldr f id xs 0
where f x g a = g (x : a)

This definition should have better time properties because, as shown above, the
steps required in evaluating the append function is proportional to the length of
the first argument, so appending the tail of the list would be inefficient.

Reverse

reverse = Abs xs (Foldr f (Val Nil) (Var xs))
where f = Abs x (Abs xs (App (App append (Var xs))

(Cons (Var x) (Val Nil))))

steps (App reverse (Val xs)) = fst (foldrVal g (10,Nil) xs)
where g z (s, zs) = (s + 21 ∗ (lengthVal zs) + 34,

eval (App (App append (Val zs)) (Val (ConsV z Nil))) [ ])

The steps function for reverse is dependent on the steps required to perform
the append function for each element, which is proportional to length of the first
argument to append . The size of this argument is increased by one each time, so
the function is a summation up to the length of the list,

8 +
length xs−1∑

x=0

21x+ 34 = 8 +
(length xs)(47 + 21(length xs))

2
6 c (length xs)2

The summation is equivalent to the fraction in the middle (from Gauss’ sum of a
finite series) which is less than a constant multiplied by the square of the length
of the list, for example, when c = 11 for all lists of length greater than 47. This
means that the function’s time requirements are quadratic on the length of the
list [9].

Reverse with an Accumulator The version of reverse with an accumulator
is expressed in the language as,

reverseAcc = Abs xs (App (Foldr f idExpr (Var xs)) (Val (Const 0)))
where f = Abs x (Abs g (Abs a (App (Var g) (Cons (Var x) (Var a)))))

The steps function is proportional to the length of the list.

stpes (App reverseAcc (Val xs)) = 26 ∗ (lengthVal xs) + 15



7 Conclusion and Further Work

We have outlined a process that takes an evaluator for a language, with a given
evaluation strategy, and derives a function that will give an accurate count of
how many steps are required to evaluate expressions in the language. The deriv-
ation itself is language and evaluation strategy non-specific. Using an extended
lambda-calculus under call-by-value evaluation, the examples in the previous
section give the appropriate complexity results that would be expected, but also
show the constants involved.

This is useful to know because of the additional overheads that functions
of the same complexity may have — for example in summing a list with and
without an accumulator. They also show the boundaries at which one function
with a lower growth rate but larger constants becomes quicker than another, for
example, in the reverse function with or without an accumulator.

This work could be taken in a few directions. Looking at what happens with
more complicated evaluation strategies, such as call-by-need/lazy evaluation,
would be one next stage. It would also be interesting to apply the same technique
to look at the space requirements for functions, as mentioned briefly earlier. A
useful addition to this work would be to develop a calculus to automate deriving
the step functions.

